Page 1



3GPP TSG-RAN5 Meeting #2018
R5s180418
Online, 15th Dec 2017,  - 31st Dec 2018
	CR-Form-v11.2.1

	CHANGE REQUEST

	

	
	36.523-3
	CR
	4219
	rev
	-
	Current version:
	15.0.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	


	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	


	

	Title:

	Correction to NB-IoT TC 22.2.5

	
	

	Source to WG:
	ROHDE & SCHWARZ

	Source to TSG:
	R5

	
	

	Work item code:
	TEI13_Test, NB_IOT-UEConTest
	
	Date:
	2018-06-22

	
	
	
	
	

	Category:
	F
	
	Release:
	Rel-15

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)

Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8
(Release 8)
Rel-9
(Release 9)
Rel-10
(Release 10)
Rel-11
(Release 11)
Rel-12
(Release 12)
Rel-13
(Release 13)
Rel-14
(Release 14)
Rel-15
(Release 15)
Rel-16
(Release 16)

	
	

	Reason for change:
	In the current TTCN implementation, at Step 27-28, the SIB4 of Ncell1 and Ncell 2 is updated. Also SIB3 for Ncell 1 is updated. This is followed by the call to the function f_NBIOT_ModifySysinfo.

In this function, the updated sysinfo is activated by calculating the next modification period. 

However, the timer for the UE to read the updated sysinfo, implemented by function f_Delay(2.1*v_ModificationPeriod_s) is started immediately.

This can result in a situation, when the activation time is around 4 to 7 HSFNs after the above function has started. Which means, that the actual updated SIBs will be broadcast when the timer of 86s, has been running for 40s or 70s already.

This would give the UE a small opportunity of 30s or less to read the updated SIBs, which means that there will be cases when the UEs may not read the updated SIBs, and the timer of 86s has expired. This will result in testcase failures.

Similar situation arises at Step 31-32 when the SIB4 is updated, and a timer of 86s is started. 

And similar situation occurs when before Step 34, when the SIB3 for Ncell 2 and Ncell 1 are updated and the power level changed.

This needs to be corrected.

As a result of the above mentioned changes, the guard timer also needs to be extended further.

	
	

	Summary of change:
	1) Modified the function f_NBIOT_ModifySysinfo. So that it returns the difference between the current time and activation time. The implementation is done in such a way that other testcases are not impacted by this change.

2) Based upon the value returned by the updated f_NBIOT_ModifySysinfo, the time gap between activation time and current time is calculated, and this is added to the 86s timer so that the UE actually gets 86s after activation to read the updated SIBs.

3) Increased the guard timer to 1800s.

	
	

	Consequences if not approved:
	A conformant UE may fail the test case

	
	

	Clauses affected:
	22.2.5

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications

	TS/TR ... CR ... 

	affected:
	
	X
	 Test specifications
	TS/TR ... CR ... 

	(show related CRs)
	
	X
	 O&M Specifications
	TS/TR ... CR ... 

	
	

	Other comments:
	


Table of Contents
3Table of Contents

1
Corrections required
4
1.1
f_NBIOT_ModifySysinfo
4
1.2
f_TC_22_2_5
6
1.3
TC_22_2_5
19



1 Corrections required

1.1 f_NBIOT_ModifySysinfo
	Function name
	f_NBIOT_ModifySysinfo

	Reason for change
	In this function, the updated sysinfo is activated by calculating the next modification period. 

However, in the testcases that use this function (22.2.5 as an example, please refer to main description) the timer for the UE to read the updated sysinfo, implemented by function f_Delay(2.1*v_ModificationPeriod_s) is started immediately.

This can result in a situation, when the activation time is around 4 to 7 HSFNs after the timer for SIB reading is started (approx 86s in case of 22.2.5). 

This would effectively means, that the actual updated SIBs will be broadcast when the timer of 86s, has been running for 40s or 70s already.

To avoid this situation, it is proposed that this function is able to return the time difference (in terms of HSFNs) to main test where it is being called so that the timers allocated for SIB reading can be adjusted accordingly.



	Summary of change
	Modified the function f_NBIOT_ModifySysinfo so that it returns the difference between the current time and activation time. The implementation is done in such a way that other testcases are not impacted by this change.
Basically, added a new boolean parameter p_ReturnNectModPeriodTiming, which if set to TRUE would result in this function returning the difference between current time and activation time in terms of HSFNs.

Please see screenshot.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	Not Needed


Before Change:

	function f_NBIOT_ModifySysinfo(NBIOT_CellId_Type p_CellId,
                                 boolean p_PagingFlag := true,
                                 boolean p_ChangeValueTag := true) runs on NBIOT_PTC return SubFrameTiming_Type
  {
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_TimeOfPaging;
    var SubFrameTiming_Type v_StartNextModPeriod;
    var integer v_ModificationPeriodFrames := f_NB_CalculateModificationPeriod (p_CellId);
    var integer v_Timing_SFN;  //Variable used to perform SFN calculation on 20 bits (HSFN*1024 + SFN)
    //Get the current SFN and subframe
    v_Timing := f_NBIOT_GetCurrentTiming(p_CellId);
    //Add processing delay to current SFN and calculate full 20 bits SFN (HSFN*1024 + SFN)
    v_Timing_SFN := v_Timing.HSFN.Number * 1024 + v_Timing.SFN.Number + tsc_SfnDelay; // @sic R5s170910 sic@
    // Calculate the next Modification Period @sic R5s170910 sic@
    v_Timing_SFN := v_Timing_SFN + (v_ModificationPeriodFrames - (v_Timing_SFN mod v_ModificationPeriodFrames));
    v_TimeOfPaging.HSFN.Number := (v_Timing_SFN / 1024) mod 1024;
    v_TimeOfPaging.SFN.Number  := v_Timing_SFN mod 1024;
    v_TimeOfPaging.Subframe.Number  := 0;
    v_Timing_SFN := v_Timing_SFN + v_ModificationPeriodFrames;
    v_StartNextModPeriod.HSFN.Number := (v_Timing_SFN / 1024) mod 1024;
    v_StartNextModPeriod.SFN.Number  := v_Timing_SFN mod 1024;
    v_StartNextModPeriod.Subframe.Number  := 0;
    if (p_PagingFlag) {
      f_NBIOT_UE_PageSysinfoMod (p_CellId, v_TimeOfPaging);
    }
    f_NBIOT_SS_ConfigureSysinfo(p_CellId, p_ChangeValueTag, cs_TimingInfo_SFN(v_StartNextModPeriod.SFN.Number, v_StartNextModPeriod.HSFN.Number));
    return v_TimeOfPaging;
  }


After Change:

	function f_NBIOT_ModifySysinfo(NBIOT_CellId_Type p_CellId,
                                 boolean p_PagingFlag := true,
                                 boolean p_ChangeValueTag := true,
                                 boolean p_ReturnNextModPeriodTiming := false /*WA#22_2_5*/) runs on NBIOT_PTC return SubFrameTiming_Type
  {
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_TimeOfPaging;
    var SubFrameTiming_Type v_StartNextModPeriod;
    var integer v_ModificationPeriodFrames := f_NB_CalculateModificationPeriod (p_CellId);
    var integer v_Timing_SFN;  //Variable used to perform SFN calculation on 20 bits (HSFN*1024 + SFN)
    //Get the current SFN and subframe
    v_Timing := f_NBIOT_GetCurrentTiming(p_CellId);
    //Add processing delay to current SFN and calculate full 20 bits SFN (HSFN*1024 + SFN)
    v_Timing_SFN := v_Timing.HSFN.Number * 1024 + v_Timing.SFN.Number + tsc_SfnDelay; // @sic R5s170910 sic@
    // Calculate the next Modification Period @sic R5s170910 sic@
    v_Timing_SFN := v_Timing_SFN + (v_ModificationPeriodFrames - (v_Timing_SFN mod v_ModificationPeriodFrames));
    v_TimeOfPaging.HSFN.Number := (v_Timing_SFN / 1024) mod 1024;
    v_TimeOfPaging.SFN.Number  := v_Timing_SFN mod 1024;
    v_TimeOfPaging.Subframe.Number  := 0;
    v_Timing_SFN := v_Timing_SFN + v_ModificationPeriodFrames;
    v_StartNextModPeriod.HSFN.Number := (v_Timing_SFN / 1024) mod 1024;
    v_StartNextModPeriod.SFN.Number  := v_Timing_SFN mod 1024;
    v_StartNextModPeriod.Subframe.Number  := 0;
    if (p_PagingFlag) {
      f_NBIOT_UE_PageSysinfoMod (p_CellId, v_TimeOfPaging);
    }
    f_NBIOT_SS_ConfigureSysinfo(p_CellId, p_ChangeValueTag, cs_TimingInfo_SFN(v_StartNextModPeriod.SFN.Number, v_StartNextModPeriod.HSFN.Number));
    //return v_TimeOfPaging;
    //WA#22_2_5 : Added new if-else
    if (p_ReturnNextModPeriodTiming)
    {
    
v_StartNextModPeriod.HSFN.Number := v_StartNextModPeriod.HSFN.Number - v_Timing.HSFN.Number;    

    
return v_StartNextModPeriod;
    }else
    {
    
return v_TimeOfPaging;
    }
  }


1.2 f_TC_22_2_5
	Function name
	f_TC_22_2_5

	Reason for change
	In the current TTCN implementation, at Step 27-28, the SIB4 of Ncell1 and Ncell 2 is updated. Also SIB3 for Ncell 1 is updated. This is followed by the call to the function f_NBIOT_ModifySysinfo.

In this function, the updated sysinfo is activated by calculating the next modification period. 

However, the timer for the UE to read the updated sysinfo, implemented by function f_Delay(2.1*v_ModificationPeriod_s) is started immediately.

This can result in a situation, when the activation time is around 4 to 7 HSFNs after the above function has started. Which means, that the actual updated SIBs will be broadcast when the timer of 86s, has been running for 40s or 70s already.

This would give the UE a small opportunity of 30s or less to read the updated SIBs, which means that there will be cases when the UEs may not read the updated SIBs, and the timer of 86s has expired. This will result in testcase failures.

Similar situation arises at Step 31-32 when the SIB4 is updated, and a timer of 86s is started. 

This needs to be corrected.

	Summary of change
	Based upon the value returned by the updated f_NBIOT_ModifySysinfo (Please see change 1.1 above), the time gap between activation time and current time is calculated, and this is added to the 86s timer so that the UE actually gets 86s after activation to read the updated SIBs.

In order to implement this, created two new variables v_ActivationGap and v_StartNextModPeriod, and used these as shown below.

Please see screenshot below (Yellow is the proposed implementation, green is old implementation which is now commented outt)

	TTCN module
	NBIOT_Idle_CellReselection.ttcn

	MCC160 Comment
	Implemented differently: Call f_NBIOT_SS_ConfigureSysinfo instead of f_NBIOT_ModifySysinfo for the cells on which the UE is not camped before steps 29 and 34


Before Change:

	function f_TC_22_2_5_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / Cell reselection / Qhyst, Qoffset, Treselection and Cell-specific reselection parameters */
    var float v_ModificationPeriod_s; // Modification period in second
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1T2;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT4T5;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT6;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT7;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT8;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT9;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT10;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT11;
    var SystemInformationBlockType3_NB_r13 v_SIB3_Cell1;
    var SystemInformationBlockType3_NB_r13 v_SIB3_Cell2;
    timer t_ResponseTimer := 26.0;
    v_CellPowerList_AtT1T2 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT3 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT4T5 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT6 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT7 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT8 := {
      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, -85)
    };
    v_CellPowerList_AtT9 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT10 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT11 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -79),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    //Initialise all Ncells, security and mobile parameters
    f_NBIOT_Init(c2);
    //Get ModificationPeriod
    v_ModificationPeriod_s := int2float(f_NB_CalculateModificationPeriod (nbiot_Cell1))/100.0;
    //Set TAC according to 36.508 Table 8.1.4.2-6: Tracking Area Code (TAC) for NBIOT cells */
    f_NBIOT_CellInfo_SetTAC(nbiot_Cell2, tsc_Idle_TAC_Cell2);
    f_NBIOT_CellInfo_SetTAC(nbiot_Cell4, tsc_Idle_TAC_Cell4);
    //Set SIB3-NB according to specific message contents(Table 22.2.5.3.3-2)
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.cellReselectionInfoCommon_r13.q_Hyst_r13 := dB24;
    f_NBIOT_CellInfo_SetSIB3 (nbiot_Cell1, v_SIB3_Cell1);
    //Set SIB4-NB according to specific message contents(Table 22.2.5.3.3-3)
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_508_Cell1_DefSIB4_NB_IntraFreqCellList);
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_AllParameters({cs_IntraFreqNeighCellInfo(0, dB0),  //q-OffsetCell = 0 dB for cell 1 @sic R5s170701 sic@
                                                                    cs_IntraFreqNeighCellInfo(4, dB0),  //q-OffsetCell = 0 dB for cell 4
                                                                    cs_IntraFreqNeighCellInfo(11, dB0)},//q-OffsetCell = 0 dB for cell 11
                                                                   { cs_PhysCellIdRange(f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell4), omit) }));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell4, cs_508_Cell4_DefSIB4_NB_IntraFreqCellList);
    //Set maximum cell power level for Ncell 2
    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell2, -79);// @sic R5s170701 sic@
    //Create and configure all NBIOT Cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell4, CONTROL_PLANE);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);
    f_NBIOT_TestBody_Set(true);
    //@siclog "Step 0" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 1" siclog@
    //Change Ncell settings according to T1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1T2);
    //@siclog "Step 2" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 2
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 10.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 1 within 10s
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 22.2.5 Step 2 : RRCConnectionRequest-NB message received on NCell 2 within wait time");
    }
    //@siclog "Step 3" siclog@
    //Change cell settings according to T2
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.cellReselectionInfoCommon_r13.q_Hyst_r13 := dB0;
    f_NBIOT_CellInfo_SetSIB3 (nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT1T2);
    //@siclog "Step 4" siclog@
    f_NBIOT_ModifySysinfo (nbiot_Cell1, true);
    //@siclog "Step 5" siclog@
    //Check for camp on Ncell 2
    f_NBIOT_508CheckCampOnNewNBIOTCell (nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 6-7" siclog@
    //Change system information according to T3
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_ThreeNeighCellEntries(2, dB24,//q-OffsetCell = 24 dB for cell 2
                                                                           4, dB0,//q-OffsetCell = 0 dB for cell 4
                                                                           11, dB0));//q-OffsetCell = 0 dB for cell 11
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    //@siclog "Step 8" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 8A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T3" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT3);
    //@siclog "Step 9" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell (nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 10" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 11" siclog@
    //Change Ncell settings according to T4
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT4T5);
    //@siclog "Step 12" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 2
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 10.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Cell 1 within 10s
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 12 : RRCConnectionRequest-NB message received on NCell 2 within wait time");
    }
    //@siclog "Step 13-14" siclog@
    //Change cell settings according to T5
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_508_Cell1_DefSIB4_NB_IntraFreqCellList);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    //@siclog "Step 15" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 16-17" siclog@
    //Change cell settings according to T6
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.t_Reselection_r13 := s21;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    //@siclog "Step 18" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 18A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T6" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT6);
    //@siclog "Step 19" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 20" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 21" siclog@
    //Change cell settings according to T7
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT7);
    //@siclog "Step 22" siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 20.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 2 within 20s
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 22 : RRCConnectionRequest-NB message received on Ncell 2 within wait time");
    }
    //@siclog "Step 23" siclog@
    //Receive RRCConnectionRequest-NB on Ncell 2 within t_ResponseTimer
    t_ResponseTimer.start;
    alt {
      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell2, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any, ?))))
         {
          //* @verdict pass RRCConnectionRequest-NB message received within t_ResponseTimer
          f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Test Case 22.2.5 Step 23");
          t_ResponseTimer.stop;// @sic R5s170701 sic@
         }
       [] t_ResponseTimer.timeout
       {
         //* @verdict fail No RRCConnectionRequest-NB message received on Ncell 2 within t_ResponseTimer time
         f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 23");
       }
    }
    //@siclog "Step 24" siclog@
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 25" siclog@
    //Change cell settings according to T8
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT8);
    //@siclog "Step 26" siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell4, 120.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 4 within 120s
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 22.2.5 Step 26 : RRCConnectionRequest-NB message received on Ncell 6 within wait time");
    }
    //@siclog "Step 27-28" siclog@
    //Change system information according to T9
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo(2, dB24)}));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_AllParameters ({cs_IntraFreqNeighCellInfo (0, dB_24)},
                                                                     { cs_PhysCellIdRange(f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell4), omit) }));// @sic R5s170701 sic@
    //@sic R5-174648 sic@
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.t_Reselection_r13 := s0;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    //@siclog "Step 29" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 29A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T9" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT9);
    //@siclog "Step 30" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 31-32" siclog@
    //Change system information according to T10
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT10);
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (2, dB0)}));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (0, dB0)}));// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    //@siclog "Step 33" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 33A-33B" siclog@
    //@sic R5-174648 sic@ Set SIB3-NB according to table 22.2.5.3.3-14
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    v_SIB3_Cell2 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell2);
    v_SIB3_Cell2.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell2, v_SIB3_Cell2);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    f_NBIOT_ModifySysinfo(nbiot_Cell2, false);
    //@siclog "Step 34" siclog@
    //Change system information according to T11
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT11);
    //@siclog "Step 35" siclog@
    //Check for camp on Ncell 2
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Test Case 22.2.5 Step 35");// @sic R5s170701 sic@
    f_NBIOT_TestBody_Set(false);
    //Switch/power off UE
    f_NBIOT_Postamble(nbiot_Cell2, CONTROL_PLANE, N1_IDLE);
  }


After Change:

	function f_TC_22_2_5_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / Cell reselection / Qhyst, Qoffset, Treselection and Cell-specific reselection parameters */
    var float v_ModificationPeriod_s; // Modification period in second
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1T2;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT4T5;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT6;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT7;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT8;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT9;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT10;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT11;
    var SystemInformationBlockType3_NB_r13 v_SIB3_Cell1;
    var SystemInformationBlockType3_NB_r13 v_SIB3_Cell2;
    timer t_ResponseTimer := 26.0;
    var SubFrameTiming_Type v_StartNextModPeriod; //WA#22_2_5
    var float v_ActivationGap; //WA#22_2_5
    v_CellPowerList_AtT1T2 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT3 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT4T5 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT6 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT7 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT8 := {
      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell2, -91),
      cs_NB_CellPower(nbiot_Cell4, -85)
    };
    v_CellPowerList_AtT9 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -85),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT10 := {
      cs_NB_CellPower(nbiot_Cell1, -85),
      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    v_CellPowerList_AtT11 := {
      cs_NB_CellPower(nbiot_Cell1, -91),
      cs_NB_CellPower(nbiot_Cell2, -79),
      cs_NB_CellPower(nbiot_Cell4, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    //Initialise all Ncells, security and mobile parameters
    f_NBIOT_Init(c2);
    //Get ModificationPeriod
    v_ModificationPeriod_s := int2float(f_NB_CalculateModificationPeriod (nbiot_Cell1))/100.0;
    //Set TAC according to 36.508 Table 8.1.4.2-6: Tracking Area Code (TAC) for NBIOT cells */
    f_NBIOT_CellInfo_SetTAC(nbiot_Cell2, tsc_Idle_TAC_Cell2);
    f_NBIOT_CellInfo_SetTAC(nbiot_Cell4, tsc_Idle_TAC_Cell4);
    //Set SIB3-NB according to specific message contents(Table 22.2.5.3.3-2)
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.cellReselectionInfoCommon_r13.q_Hyst_r13 := dB24;
    f_NBIOT_CellInfo_SetSIB3 (nbiot_Cell1, v_SIB3_Cell1);
    //Set SIB4-NB according to specific message contents(Table 22.2.5.3.3-3)
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_508_Cell1_DefSIB4_NB_IntraFreqCellList);
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_AllParameters({cs_IntraFreqNeighCellInfo(0, dB0),  //q-OffsetCell = 0 dB for cell 1 @sic R5s170701 sic@
                                                                    cs_IntraFreqNeighCellInfo(4, dB0),  //q-OffsetCell = 0 dB for cell 4
                                                                    cs_IntraFreqNeighCellInfo(11, dB0)},//q-OffsetCell = 0 dB for cell 11
                                                                   { cs_PhysCellIdRange(f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell4), omit) }));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell4, cs_508_Cell4_DefSIB4_NB_IntraFreqCellList);
    //Set maximum cell power level for Ncell 2
    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell2, -79);// @sic R5s170701 sic@
    //Create and configure all NBIOT Cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell4, CONTROL_PLANE);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);
    f_NBIOT_TestBody_Set(true);
    //@siclog "Step 0" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 1" siclog@
    //Change Ncell settings according to T1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1T2);
    //@siclog "Step 2" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 2
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 10.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 1 within 10s
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 22.2.5 Step 2 : RRCConnectionRequest-NB message received on NCell 2 within wait time");
    }
    //@siclog "Step 3" siclog@
    //Change cell settings according to T2
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.cellReselectionInfoCommon_r13.q_Hyst_r13 := dB0;
    f_NBIOT_CellInfo_SetSIB3 (nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT1T2);
    //@siclog "Step 4" siclog@
    f_NBIOT_ModifySysinfo (nbiot_Cell1, true);
    //@siclog "Step 5" siclog@
    //Check for camp on Ncell 2
    f_NBIOT_508CheckCampOnNewNBIOTCell (nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 6-7" siclog@
    //Change system information according to T3
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_ThreeNeighCellEntries(2, dB24,//q-OffsetCell = 24 dB for cell 2
                                                                           4, dB0,//q-OffsetCell = 0 dB for cell 4
                                                                           11, dB0));//q-OffsetCell = 0 dB for cell 11
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    //@siclog "Step 8" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 8A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T3" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT3);
    //@siclog "Step 9" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell (nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 10" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 11" siclog@
    //Change Ncell settings according to T4
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT4T5);
    //@siclog "Step 12" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 2
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 10.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Cell 1 within 10s
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 12 : RRCConnectionRequest-NB message received on NCell 2 within wait time");
    }
    //@siclog "Step 13-14" siclog@
    //Change cell settings according to T5
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_508_Cell1_DefSIB4_NB_IntraFreqCellList);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    //@siclog "Step 15" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 16-17" siclog@
    //Change cell settings according to T6
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.t_Reselection_r13 := s21;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    //@siclog "Step 18" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    f_Delay(2.1 * v_ModificationPeriod_s);
    //@siclog "Step 18A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T6" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT6);
    //@siclog "Step 19" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 20" siclog@
    //Wait for 1 sec after preamble to allow UE to come to idle mode
    f_Delay(1.0);
    //@siclog "Step 21" siclog@
    //Change cell settings according to T7
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT7);
    //@siclog "Step 22" siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell2, 20.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 2 within 20s
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 22 : RRCConnectionRequest-NB message received on Ncell 2 within wait time");
    }
    //@siclog "Step 23" siclog@
    //Receive RRCConnectionRequest-NB on Ncell 2 within t_ResponseTimer
    t_ResponseTimer.start;
    alt {
      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell2, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any, ?))))
         {
          //* @verdict pass RRCConnectionRequest-NB message received within t_ResponseTimer
          f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Test Case 22.2.5 Step 23");
          t_ResponseTimer.stop;// @sic R5s170701 sic@
         }
       [] t_ResponseTimer.timeout
       {
         //* @verdict fail No RRCConnectionRequest-NB message received on Ncell 2 within t_ResponseTimer time
         f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 22.2.5 Step 23");
       }
    }
    //@siclog "Step 24" siclog@
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    //@siclog "Step 25" siclog@
    //Change cell settings according to T8
    f_NBIOT_SetCellPowerList (v_CellPowerList_AtT8);
    //@siclog "Step 26" siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell4, 120.0)) {
      //* @verdict fail UE sent RRCConnectionRequest-NB message on Ncell 4 within 120s
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 22.2.5 Step 26 : RRCConnectionRequest-NB message received on Ncell 6 within wait time");
    }
    //@siclog "Step 27-28" siclog@
    //Change system information according to T9
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo(2, dB24)}));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_AllParameters ({cs_IntraFreqNeighCellInfo (0, dB_24)},
                                                                     { cs_PhysCellIdRange(f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell4), omit) }));// @sic R5s170701 sic@
    //@sic R5-174648 sic@
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.t_Reselection_r13 := s0;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    f_NBIOT_ModifySysinfo(nbiot_Cell1, false);
    //WA#22_2_5 f_NBIOT_ModifySysinfo(nbiot_Cell2, true);
    v_StartNextModPeriod := f_NBIOT_ModifySysinfo(nbiot_Cell2, true, -, true);//WA#22_2_5    
    v_ActivationGap := int2float ((v_StartNextModPeriod.HSFN.Number*10*1000 + v_StartNextModPeriod.SFN.Number*10)/1000); //WA#22_2_5
    //@siclog "Step 29" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    //WA#22_2_5 f_Delay(2.1 * v_ModificationPeriod_s);
    f_Delay((2.1 * v_ModificationPeriod_s) + v_ActivationGap); //WA#22_2_5
    //@siclog "Step 29A" siclog@
    //SS re-adjusts cell-specific reference signal levels according to rows "T9" in table 22.2.5.3.2-1
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT9);
    //@siclog "Step 30" siclog@
    //Check for camp on Ncell 1
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 31-32" siclog@
    //Change system information according to T10
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT10);
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (2, dB0)}));
    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (0, dB0)}));// @sic R5s170701 sic@
    //WA#22_2_5 f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    v_StartNextModPeriod := f_NBIOT_ModifySysinfo(nbiot_Cell1, true, -, true);//WA#22_2_5    
    v_ActivationGap := int2float ((v_StartNextModPeriod.HSFN.Number*10*1000 + v_StartNextModPeriod.SFN.Number*10)/1000); //WA#22_2_5
    //@siclog "Step 33" siclog@
    //Wait 2.1*modification period to allow new system information to take effect
    //WA#22_2_5 f_Delay(2.1 * v_ModificationPeriod_s);
    f_Delay((2.1 * v_ModificationPeriod_s) + v_ActivationGap); //WA#22_2_5
    //@siclog "Step 33A-33B" siclog@
    //@sic R5-174648 sic@ Set SIB3-NB according to table 22.2.5.3.3-14
    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);
    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);
    v_SIB3_Cell2 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell2);
    v_SIB3_Cell2.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;
    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell2, v_SIB3_Cell2);// @sic R5s170701 sic@
    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);
    //WA#22_2_5 f_NBIOT_ModifySysinfo(nbiot_Cell2, false);
    v_StartNextModPeriod := f_NBIOT_ModifySysinfo(nbiot_Cell2, true, -, true);//WA#22_2_5    
    v_ActivationGap := int2float ((v_StartNextModPeriod.HSFN.Number*10*1000 + v_StartNextModPeriod.SFN.Number*10)/1000); //WA#22_2_5
    f_Delay((2.1 * v_ModificationPeriod_s) + v_ActivationGap); //WA#22_2_5
    //@siclog "Step 34" siclog@
    //Change system information according to T11
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT11);
    //@siclog "Step 35" siclog@
    //Check for camp on Ncell 2
    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell2, CONTROL_PLANE);
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Test Case 22.2.5 Step 35");// @sic R5s170701 sic@
    f_NBIOT_TestBody_Set(false);
    //Switch/power off UE
    f_NBIOT_Postamble(nbiot_Cell2, CONTROL_PLANE, N1_IDLE);
  }


MCC160 Implementation
     function f_TC_22_2_5_NBIOT() runs on NBIOT_PTC   
...

    //@siclog "Step 27-28" siclog@

    //Change system information according to T9

    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo(2, dB24)}));

    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_AllParameters ({cs_IntraFreqNeighCellInfo (0, dB_24)},

                                                                     { cs_PhysCellIdRange(f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell4), omit) }));// @sic R5s170701 sic@

    //@sic R5-174648 sic@

    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);

    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.t_Reselection_r13 := s0;

    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);

    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1);
    f_NBIOT_ModifySysinfo(nbiot_Cell2, true);

    //@siclog "Step 29" siclog@

    //Wait 2.1*modification period to allow new system information to take effect

    f_Delay(2.1 * v_ModificationPeriod_s);

    //@siclog "Step 29A" siclog@

    //SS re-adjusts cell-specific reference signal levels according to rows "T9" in table 22.2.5.3.2-1

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT9);

    //@siclog "Step 30" siclog@

    //Check for camp on Ncell 1

    f_NBIOT_508CheckCampOnNewNBIOTCell(nbiot_Cell1, CONTROL_PLANE);

    //@siclog "Step 31-32" siclog@

    //Change system information according to T10

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT10);

    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell1, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (2, dB0)}));

    f_NBIOT_CellInfo_SetSIB4(nbiot_Cell2, cs_SIB4_NB_NeighCellList({cs_IntraFreqNeighCellInfo (0, dB0)}));// @sic R5s170701 sic@

    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);

    //@siclog "Step 33" siclog@

    //Wait 2.1*modification period to allow new system information to take effect

    f_Delay(2.1 * v_ModificationPeriod_s);

    //@siclog "Step 33A-33B" siclog@

    //@sic R5-174648 sic@ Set SIB3-NB according to table 22.2.5.3.3-14

    v_SIB3_Cell1 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell1);

    v_SIB3_Cell1.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;

    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell1, v_SIB3_Cell1);

    v_SIB3_Cell2 := f_NBIOT_CellInfo_GetSIB3(nbiot_Cell2);

    v_SIB3_Cell2.intraFreqCellReselectionInfo_r13.s_IntraSearchP_r13 := 11;

    f_NBIOT_CellInfo_SetSIB3(nbiot_Cell2, v_SIB3_Cell2);// @sic R5s170701 sic@

    f_NBIOT_ModifySysinfo(nbiot_Cell1, true);

    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell2);
    //@siclog "Step 34" siclog@

    //Change system information according to T11

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT11);

    //@siclog "Step 35" siclog@
1.3 TC_22_2_5
	Function name
	TC_22_2_5

	Reason for change
	Because od the changes 1.1 and 1.2 above, the guard timer has to be increased to 1800s.

	Summary of change
	Increased the guard timer to 1800s
Please see screenshot.

	TTCN module
	NBIOT_Testsuite.ttcn

	MCC160 Comment
	Not needed


Before Change:

	testcase TC_22_2_5() runs on MTC_NBIOT system SYSTEM_NBIOT {
    // @purpose
    //   NB-IoT / Cell reselection / Qhyst, Qoffset, Treselection and Cell-specific reselection parameters
    var NBIOT_PTC        v_NBIOT      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(1200);
    v_NBIOT := NBIOT_PTC.create alive;
    f_MTC_ConnectPTCs_NBIOT(system, v_NBIOT, v_EUTRA);
    v_NBIOT.start(f_TC_22_2_5_NBIOT());
    t_GuardTimer.start;
    f_MTC_MainLoop_NBIOT(t_GuardTimer);
  }


After Change:

	testcase TC_22_2_5() runs on MTC_NBIOT system SYSTEM_NBIOT {
    // @purpose
    //   NB-IoT / Cell reselection / Qhyst, Qoffset, Treselection and Cell-specific reselection parameters
    var NBIOT_PTC        v_NBIOT      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(1800); //WA#22_2_5 1200->1800
    v_NBIOT := NBIOT_PTC.create alive;
    f_MTC_ConnectPTCs_NBIOT(system, v_NBIOT, v_EUTRA);
    v_NBIOT.start(f_TC_22_2_5_NBIOT());
    t_GuardTimer.start;
    f_MTC_MainLoop_NBIOT(t_GuardTimer);
  }


